

Summer School on Generic and Effectful Programming

**Applying Type-Level and
Generic Programming in Haskell**

Andres Löh, Well-Typed LLP

July 4, 2015

Contents

1	Introduction and type-level programming	5
1.1	Motivation	5
1.1.1	What is datatype-generic programming?	5
1.1.2	Applications of generic programming	6
1.1.3	A uniform view on data	6
1.2	Kinds and data kinds	7
1.2.1	Stars and functions	8
1.2.2	Promoted data kinds	8
1.3	Generalized algebraic data types (GADTs)	9
1.3.1	Vectors	9
1.3.2	Functions on vectors	10
1.4	Singleton types	11
1.4.1	Singleton natural numbers	11
1.4.2	Evaluation	12
1.4.3	Applicative vectors	13
1.5	Heterogeneous lists	13
1.5.1	Promoted lists and kind polymorphism	13
1.5.2	Environment or n -ary products	14
1.6	Higher-rank types	15
1.6.1	Mapping over n -ary products	15
1.6.2	Applicative n -ary products	17
1.6.3	Lifted functions	18
1.6.4	Another look at <code>hmap</code>	19
1.7	Abstracting from classes and type functions	19
1.7.1	The kind <code>Constraint</code>	20
1.7.2	Type functions	20
1.7.3	Composing constraints	21
1.7.4	Proxies	22
1.7.5	A variant of <code>hpure</code> for constrained functions	23
2	Generic programming with n-ary sums of products	25
2.1	Recap: n -ary products <code>NP</code>	25
2.2	Generalizing choice	25
2.2.1	Choosing from a list	26
2.2.2	Choosing from a vector	26
2.2.3	Choosing from a heterogeneous list	26
2.2.4	Choosing from an environment	27
2.3	Sums of products	27

2.3.1	Representing expressions	27
2.3.2	The <code>Generic</code> class	29
2.3.3	Other <code>Generic</code> instances	29
2.3.4	Support for <code>Generic</code> in the compiler	30
2.4	Defining generic equality	31
2.4.1	A bottom-up approach	31
2.4.2	The <code>All2</code> constraint	32
2.4.3	Completing the definition	32
2.4.4	Improving the product case	33

1 Introduction and type-level programming

Please note that this is a preliminary version of the lecture notes, which does not contain the material corresponding to the final parts of the lecture series, and may contain various glitches and holes. You will get an updated and completed version during the summer school. Feedback and suggestions for improvement are nevertheless welcome for any version.

1.1 Motivation

1.1.1 What is datatype-generic programming?

The idea of datatype-generic programming is as follows. Assume you have a datatype `A` and consider a function on that datatype such as structural equality:

```
eqA :: A -> A -> Bool
```

Given the definition of `A`, it's very easy and straight-forward to give the definition of `eqA`. That's because there is an algorithm that we can describe informally: if the datatype admits different choices (i.e., *constructors*), then check that the two arguments are of the same constructor. If they are, then check the arguments of the constructor for equality pointwise.

If we can phrase the algorithm informally, we'd like to also be able to write it down formally. (Haskell allows to say **deriving Eq** on a datatype declaration, invoking compiler magic that conjures up a suitable definition of equality for that datatype, but that isn't a formally defined algorithm and only works for a handful of functions.)

So here is the general idea, expressed in terms of Haskell: assume there is a type class

```
class Generic a where
  type Rep a
  from :: a -> Rep a
  to   :: Rep a -> a
```

that associates with a type `a` an isomorphic representation type `Rep a`, witnessed by the conversion functions `from` and `to`. Now if all `Rep a` types have a common structure, we can define a generic function

```
geq :: Generic a => Rep a -> Rep a -> Bool
```

that works for all representation types by induction over this common structure.

Then we can use the isomorphism and `geq` to get an equality function that works on any representable type:

```
eq :: Generic a => a -> a -> Bool
eq x y = geq (from x) (from y)
```

We can then build a limited amount of compiler magic into the language to derive the `Generic` instance for each representable type automatically, i.e., to automatically come up with a structural representation type the conversion functions, and we have reduced the need for having an arbitrary amount of magically derivable type classes to the need to have just a single magically derivable type class, called `Generic`.

There is still a lot of flexibility in the details of how exactly the class `Generic` is defined, and even more importantly, what kind of representation we choose for `Rep`. This choice is often called the generic *view* or *universe*, and it is the main way in which the countless generic programming approaches that exist for Haskell (SYB, RepLib, instant-generics, regular, multirec, generic-deriving, generic-sop) are different from one another. This choice can have a significant effect:

- Depending on the choice of universe, some datatypes may be representable and others may not.
- Depending on the choice of universe, some generic functions may be definable and others may not definable (or more difficult to define, or less efficient).

In the context of this lecture series, we are going to look at one such universe, which is relatively recent, called `generics-sop`. I think it is quite elegant and easy to use, while still being expressive. We will also briefly look at the relation to other universes, and I will take that opportunity to argue that despite the arguments I just made, the choice of universe is less important than one might think.

In order to work with `generics-sop`, one has to use quite a few type system extensions offered by GHC, and it helps to understand the concepts involved. One may see this as a disadvantage, but here, we're going to see it as a welcome challenge: being able to program on the type level in GHC, while not a silver bullet, is an exciting area in itself, and useful for many purposes beyond datatype-generic programming, so we'll see it as a useful opportunity to familiarize ourselves with the necessary concepts.

Apart from the knowledge barrier, `generics-sop` is ready for use. It is released, available on Hackage, and is being used.

1.1.2 Applications of generic programming

There are many ways in which datatype-generic programs can be usefully implied. Next to equality and comparison functions, there are a large class of translation functions that translate between values of datatypes and other formats, such as human-readable text, binary encodings or data description languages such as JSON, XML or similar. Typical generic functions include also all sorts of navigation or traversal functions that access certain parts of values while leaving others alone, such as lenses.

In the latter part of this lecture series, we will discuss a number of applications of `generics-sop`.

1.1.3 A uniform view on data

We are now going to explain the main motivation behind the structural representation of Haskell datatypes that is chosen by `generics-sop`.

Let us start by looking at a few datatypes in Haskell:

```
data Maybe a    = Nothing | Just a
data Either a b = Left a | Right b
data Group      = Group Char Bool Int
data Expr       = NumLit Int | BoolLit Bool | Add Expr Expr | If Expr Expr Expr
```

We observe: Haskell datatypes offer a *choice* between different constructors. Depending on the constructor chosen, we have to provide an appropriate *sequence* of arguments. In other words, each value of such a datatype is a constructor applied to a number of arguments. This is going to be our common structure:

$$C_i \ x_0 \dots x_{n_i-1}$$

If we know the type of the value, the names of its constructors do not really matter. We can simply number the constructors from 0 to $c - 1$ where c is the total number of constructors for that datatype. Depending on the chosen number n , we know how many arguments n_i we expect and what their types are.

Interestingly, this representation is quite close to the run-time representation of Haskell values. GHC stores each value starting with a pointer to some information that, among other things, contains a tag indicating the number of the constructor. Following this info pointer there is the “payload”, which is just a sequence of pointers to the constructor arguments.

So it is plausible that something similar might work for us. But we need to assign types to this representation if we want to work with it on the Haskell surface level. Which means that we have to formalize concepts such as “the number of the constructor determines the number of arguments and their types”.

Haskell does not have “full-spectrum” dependent types, but it is well-known that we can get very close with various GHC extensions. We will model the sequence of arguments of a constructor as a typed heterogeneous list. Then we will model the choice between the constructors as something like a pointer into a heterogeneous list. In the following, we will introduce the GHC concepts that we will rely on, such as GADTs, data kinds and constraint kinds.

Since type-level programming is a bit peculiar in Haskell, we’ll move step by step: from normal list over length-indexed vectors to heterogeneous lists and ultimately the slightly more generalized n -ary products that we are planning to use. Several concepts will be introduced in the slightly simpler setting of length-indexed vectors before we try to apply them to n -ary products.

1.2 Kinds and data kinds

Haskell has a layered type system. Types are defined using a much more limited language than terms. Perhaps the most obvious omission is the lack of a plain lambda abstraction on the type level. Nevertheless, types are structured, and there is the possibility to apply types to one another, and to parameterize named types. Therefore, types need their own, non-trivial, types, which are called kinds.

1.2.1 Stars and functions

For a long time, the kind system of Haskell was quite uninteresting:

- The kind `*` is for types that classify terms (even though not every type of kind `*` need be inhabited). In particular, whenever we define a new datatype using `data`, its fully applied form is of kind `*`. This also implies that kind `*` is *open*. New members of kind `*` can be added via `data` at any time.
- If `k` and `l` are kinds, we can form the function kind `k -> l` to indicate types of kind `l` that are parameterized over types of kind `k`.

Examples:

- The types `Int`, `Double`, `Bool`, and `Char` are all of kind `*`.
- The types `Maybe`, `[]`, and `IO` are of kind `* -> *`, but e.g. `Maybe Int`, `[Double]` or `IO Bool` are of kind `*` again.
- The types `Either` or `(,)` are of kind `* -> * -> *`. They can be partially applied, so e.g. `Either Int` is of kind `* -> *`. And `Either Int Char` or `(Bool, Double)` are of kind `*` again.
- We can use the kind system to tell that `Either (Maybe [Int]) (Char, Bool)` is well-formed, but `(Maybe, IO)` is not.

1.2.2 Promoted data kinds

Nowadays, Haskell has many more kinds than just the above. In fact, we can define new kinds ourselves, via datatype promotion. Whenever we define a new datatype, we can *also* use it as a kind.

Example:

```
data Bool = False | True
```

This defines the datatype `Bool` with (data) constructors

```
False :: Bool  
True  :: Bool
```

But promotion allows us to use everything also one level up, so `Bool` is a new *kind* with *type constructors*

```
'False :: Bool  
'True  :: Bool
```

Just like there are only two values of type `Bool` (ignoring undefined or “bottom”), there are only two types of kind `Bool`. The quotes can be used to emphasize the promotion and are required where the code would otherwise be syntactically ambiguous, but are often just dropped.

So if `'False` is a type, are there any terms or values of type `'False`? The answer is no! The kind `*` remains the kind of types that classify values. All the other new kinds we now gain by promoting datatypes are uninhabited.

They are, however, still useful, as we shall see, because they can appear as the arguments and results of type-level functions, and they can appear as parameters of datatypes and classes. It's the latter use that we are for now most interested in.

1.3 Generalized algebraic data types (GADTs)

We'd like to model heterogeneous lists so that we can represent sequences of arguments of a data constructor as such lists. We'll do so by generalizing normal Haskell lists in several steps.

1.3.1 Vectors

The first step is to go to length-indexed lists, also known as *vectors*. For these, let's first define the natural numbers as a datatype as follows:

```
data Nat = Zero | Suc Nat
```

We are going to use this datatype in promoted form, so are going to see `Nat` as a kind, and

```
'Zero :: Nat
'Suc  :: Nat -> Nat
```

(again, the quotes are generally optional) as types. The definition of vectors is then as follows:

```
data Vec (a :: *) (n :: Nat) where
  VNil  :: Vec a Zero
  VCons :: a -> Vec a n -> Vec a (Suc n)
infixr 5 'VCons'
```

This definition uses a different, more flexible syntax for datatype definitions: we define `Vec :: * -> Nat -> *` with two constructors, by listing their types. This syntax gives us the option to restrict the result type. For example, `Nil` does not construct an arbitrary vector `Vec a n`, but only a `Vec a Zero`. Similarly, `Cons` cannot be used to construct a `Vec a Zero`, but only a `Vec a (Suc n)` for some `n`.

Note the role of `Nat` here, as just an index. As we have observed above, there are no inhabitants of type `Nat`, nor is there any need. We just use `Nat` to grant us more information about vectors. In a full-spectrum dependently typed language, we could use the *actual* natural numbers for this. In Haskell, we have to lift them to the type level, but apart from that technicality, the effect is very similar.

Let us look at an example:

```
type Three = Suc (Suc (Suc Zero))
vabc :: Vec Char Three
vabc = 'a' 'VCons' 'b' 'VCons' 'c' 'VCons' VNil
```

The type of `vabc` can also be inferred. The type system keeps track of the length of our vectors, as desired.

It's a bit annoying if we cannot show vectors in GHCi, so let's add a `deriving` declaration that magically gives us a `Show` instance for vectors:

```
deriving instance Show a => Show (Vec a n)
```

The true power of GADTs becomes visible if we consider pattern matching, such as in

```
vtail :: Vec a (Suc n) -> Vec a n
vtail (VCons x xs) = xs
```

or

```
vmap :: (a -> b) -> Vec a n -> Vec b n
vmap f VNil      = VNil
vmap f (VCons x xs) = f x `VCons` vmap f xs
```

In `vtail`, the type signature indicates that we receive a non-empty vector. This means we do not have to match on `Nil` – in fact, doing so would not be type-correct. We can write a safe and total `vtail` function.

In `vmap`, matching on `Nil` tells GHC that for this case `n` must be equal to `Zero`. GHC has its own syntax for type equality: it will learn that `n ~ Zero` holds for this case, and therefore, it will happily accept that we return `Nil` again. Trying to use `Cons` on the right hand side of the `Nil` case would cause a type error.

Similarly, in the `Cons`-case, GHC learns `n ~ Suc n'` for some `n'`. Then, because `vmap` preserves the length, `vmap f xs :: Vec a n'` and `Cons`-ing one element on that will yield a vector of length `n` again.

The type signatures for `vtail` and `vmap` are required. In general, pattern matching on a GADT requires an explicit type annotation.

1.3.2 Functions on vectors

Before we move on to heterogeneous lists, let's spend a bit more time trying to reimplement some functions we know from normal lists on vectors. Ultimately, our goal is to do the same for heterogeneous lists, but we'll need a couple of new concepts along the way.

Some of these concepts are needed just generally because we move more information to the type level. Others result from the further generalization from vectors to heterogeneous lists. Looking at vectors in more detail now allows us to separate these issues, rather than having to discuss and understand them all at the same time.

Consider `replicate`:

```
replicate :: Int -> a -> [a]
replicate n x
  | n <= 0    = []
  | otherwise = x : replicate (n - 1) x
```

A call to `replicate n x` creates a list with `n` copies of `x`. How would we do something similar for vectors?

The size of a vector is statically known, so we'll have to assume that the number of copies we want is also known to us at the type level. But how do we pass it? We certainly cannot use an `Int`, because that would live at the term level. Do we need an argument at all? Might we be able to write:

```
vreplicate :: a -> Vec a n
```

Unfortunately, no. This type suggests that the function is parametric in `n`, i.e., that we do not need to look at `n` at all. But we have to make a compile-time distinction based on the type `n`. One option is to use a type class:

```
class VReplicate (n :: Nat) where
  vreplicate :: a -> Vec a n
```

Haskell type classes naturally extend to the presence of data kinds.

Now it's easy to complete the definition of `VReplicate`:

```
instance VReplicate Zero where
  vreplicate _ = VNil
instance VReplicate n => VReplicate (Suc n) where
  vreplicate x = x `VCons` vreplicate x
```

This works. For example

```
GHCi> vreplicate 'x' :: Vec Char Three
VCons 'x' (VCons 'x' (VCons 'x' VNil))
```

While using type classes like this is a possibility, there also is a significant disadvantage: when following this pattern, we have to define new type classes for a lot of different functions, and the class constraints will appear in the types of all the functions that use them. In other words, we will leak information that should be internal to the implementation of a function in the types. Once we start optimizing a function or changing the algorithm or reimplementing it in terms of other, more general functions, we might have to adapt its type and therefore lots of other types in our program.

It's therefore worth to consider another option.

1.4 Singleton types

1.4.1 Singleton natural numbers

We can try to mirror the type of `replicate` more closely by creating a value that does nothing more than to help GHC to know what value of `n` we want. The idea is that we create a GADT `SNat n` such that `SNat n` has exactly one value for each `n`. Then we can use pattern matching on the `SNat n` to learn at run-time what `n` is, and branch on that.

One common approach to achieve this goal is to define:

```
data SNat (n :: Nat) where -- preliminary
  SZero :: SNat Zero
  SSuc  :: SNat n -> SNat (Suc n)
```

For example,

```
sThree :: SNat Three
sThree = SSuc (SSuc (SSuc SZero))
```

However, we're going to take a slightly different approach here by defining `SNat` and a class that are mutually recursive:

```
data SNat (n :: Nat) where
  SZero :: SNat Zero
  SSuc  :: SNatI n => SNat (Suc n)

class SNatI (n :: Nat) where
  sNat :: SNat n

instance SNatI Zero where
  sNat = SZero

instance SNatI n => SNatI (Suc n) where
  sNat = SSuc
```

Using `SNat` and `SNatI`, we can define `vreplicate` also as follows:

```
vreplicate :: forall a n . SNatI n => a -> Vec a n
vreplicate x = go sNat
  where
    go :: forall n . SNat n -> Vec a n
    go SZero = VNil
    go SSuc  = x `VCons` go sNat
```

1.4.2 Evaluation

Comparing to the original solution of defining one class per function, the singletons approach has the advantage that we need only one class per type that we pattern match on – and we do not even necessarily need the class, if we're willing to pass the singleton explicitly instead. So this gives us rather more interface stability and exposes less of the implementation. There's also a disadvantage though: type class resolution happens at compile time. However, the value of type `SNat` is an actual run-time value, and pattern matching on it happens at run-time. So the singleton-based version of `replicateS` does unnecessary work at run-time by perming case distinctions for which the outcome was already known at compile time.

A sufficiently optimizing compiler could in principle resolve all this and compile both versions to the same code. In practice, GHC does not, and while this is somewhat annoying, it can in some circumstances also be beneficial, because it actually provides the programmer with a (albeit rather subtle) choice: use type classes and get compile-time resolution (which is usually more efficient, but can actually in extreme cases produce lots of code), or use singleton and get run-time resolution (which usually does unnecessary work at run-time, but can be more compact).

1.4.3 Applicative vectors

Using `vreplicate`, vectors in principle admit the interface of an applicative functor:

```
class Functor f => Applicative f where
  pure  :: a -> f a
  (<*>) :: f (a -> b) -> f a -> f b
```

It is well known that one possible instance for lists is that where `pure` produces an infinite list of copies, and `<*>` zips together the list, applying the list of functions pointwise to the list of arguments.

For vectors, we can implement the same idea, only that of course, the length of all the lists involved is statically fixed. The function `vreplicate` plays the role of `pure`, and

```
vapply :: Vec (a -> b) n -> Vec a n -> Vec b n
vapply VNil          VNil          = VNil
vapply (f `VCons` fs) (x `VCons` xs) = f x `VCons` vapply fs xs
```

is like `<*>`.

There's a relatively technical reason why we cannot make `Vec` an actual instance of `Applicative`: The parameters of `Vec` are in the wrong order for that.

1.5 Heterogeneous lists

A heterogeneous list is like a vector: a list-like structure with a statically known length. But in addition, we also know the type of each element! So indexing by a natural number is not enough. We have to know all the types of all the elements. So we index by a normal list of types.

1.5.1 Promoted lists and kind polymorphism

Normal term-level Haskell lists define a type constructor `[]` of kind `* -> *`, with constructors

```
[] :: [a]
(:) :: a -> [a] -> [a]
```

Just like we can promote types like `Bool` and `Nat`, we can also promote lists and treat `[]` as a *kind constructor*, and `[]` and `(:)` as types.

Note that even if seen as a type constructor, `(:)` has kind

```
(:) :: a -> [a] -> [a]
```

This means it is *kind-polymorphic*. We can build a type-level list of promoted Booleans as well as a type-level list of promoted natural numbers:

```
GHCi> :kind [True, False]
[True, False] :: [Bool]
GHCi> :kind [Zero, Three]
[Zero, Three] :: [Nat]
```

For a heterogeneous list, we want a type-level list of types, which has kind `[*]`:

```
GHCi> :kind [Char, Bool, Int]
[Char, Bool, Int] :: [*]
```

The definition is as follows:

```
data HList (xs :: [*]) where
  HNil  :: HList '[]
  HCons :: x -> HList xs -> HList (x ': xs)
infixr 5 'HCons'
```

In this case, the quotes in `'[]` and `':` are actually required by GHC, because we e.g. need to disambiguate between the list type constructor `[] :: * -> *` and the empty type-level list `'[] :: [*]`.

Let's consider our example datatype `Group` again:

```
data Group = Group Char Bool Int
```

Using `HList`, we can produce a sequence of a character, a Boolean, and an integer, as follows:

```
group :: HList '[Char, Bool, Int]
group = 'x' 'HCons' False 'HCons' 3 'HCons' HNil
```

Once again, we include the type signature for reference, but it could have been inferred.

1.5.2 Environment or *n*-ary products

It turns out that we will often need heterogeneous lists where each element is given by applying a particular type constructor to one of the types in the index list. Consider for example a heterogeneous list where all values are optional (`Maybe`), or all values are IO actions.

We therefore are going to define the following variant of `HList`, which is additional abstracted over a type constructor `f`, and has elements of the type `f x` where `x` is a member of the index list:

```
data NP (f :: k -> *) (xs :: [k]) where
  Nil  :: NP f '[]
  (:*) :: f x -> NP f xs -> NP f (x ': xs)
infixr 5 :*
```

Here, `NP` is for *n*-ary product, but I sometimes also call this type an *environment*, and the index list `xs` its *signature*.

Note that we've made `NP` kind-polymorphic. We do not require the signature to be of kind `[*]`, but allow it to be an arbitrary type-level list of kind `[k]`, as long as the "interpretation function" `f` maps `k` back to `*`. This is possible, because elements of the signature do not directly appear in an environment, but only as arguments to `f`. It also will turn out to be useful for us in a little while, but let's first consider scenarios where `k` is `*`.

The datatype `NP` is really a generalization of `HList`:

```
newtype I a = I {unI :: a}
```

implements an identity function on types. For any choice of a the type `I a` is isomorphic to type `a`.

And the type `NP I` is isomorphic to `HList`, as is witnessed by the following conversion functions:

```
fromHList :: HList xs -> NP I xs
fromHList HNil          = Nil
fromHList (x 'HCons' xs) = I x :* fromHList xs
toHList  :: NP I xs -> HList xs
toHList Nil            = HNil
toHList (I x :* xs)   = x 'HCons' toHList xs
```

Interestingly, there is no need to put these conversion functions into type classes (such as `hshow`), because there are no constraints on the type-level list `xs`.

Perhaps surprisingly, `NP` is also a rather direct generalization of (homogeneous) vectors:

```
newtype K a b = K {unK :: a}
```

implements a constant function on types. For any choice of a and b, the type `K a b` is isomorphic to type `a`.

An environment of type

```
NP (K Double) '[Char, Bool, Int]
```

is a list of three elements of type `Double`. The length of the type-level list determines the number of elements, however the types in the index list are irrelevant, as they're all ignored by `K`.

In practice, the function that turns out to be useful often is one that collapses an `NP` of `Ks` into a normal list:

```
hcollapse :: NP (K a) xs -> [a]
hcollapse Nil = []
hcollapse (K x :* xs) = x : hcollapse xs
```

1.6 Higher-rank types

Let's try to define a number of functions on environments. For vectors, we had implemented `vmap`, `vreplicate` and `vapply`, which together gave us (morally speaking) an instance of `Applicative` for vectors. Our aim is to provide a similar interface for environments.

1.6.1 Mapping over *n*-ary products

Let's start by looking at a generalization of `vmap`.

```
vmap :: (a -> b) -> Vec a n -> Vec b n
```

Note that `vmap` preserves the length index, but changes the type of elements.

For `NP`, we have no single type of elements, but different types. While we might want to change the type of elements, we better not change it in completely unpredictable ways, otherwise we cannot do anything useful with the list anymore. Here, it turns out to be useful that we've made the step from `HList` to `NP`, because we already have a type constructor `f` that tells us how to interpret each element in the signature. So let us generalize `vmap` to `hmap` in the following way: the function `hmap` will also preserve the index, but change the type constructor.

This turns out to be very flexible in practice. For example, with just the two instantiations of `f` we have considered so far, `K` and `I`, we can model a type-preserving map as a function

```
NP I xs -> NP I xs
```

and a type-unifying map as a function

```
NP I xs -> NP (K a) xs
```

and a map that actually produces values of different types out of a homogeneous list as a function

```
NP (K a) xs -> NP I xs
```

But note that we can also plug other type constructors into an `NP`, including GADTs, so in principle we can express rather strange maps, as long as we are willing to explain them to the type system.

So let's look at the code we would expect `hmap` to have, and think about the type signature afterwards:

```
hmap m Nil      = Nil
hmap m (x :* xs) = m x :* hmap m xs
```

Ok, so `hmap` clearly takes an environment and produces one again, with equally many elements, as was the plan:

```
hmap :: ... -> NP f xs -> NP g xs
```

Now it seems that we need a function of type `f x -> g x` as an argument. But for what `x`? If you look at the second case, then `m` will be applied to all elements of the list `xs`. So it's not enough if `m` works for one particular `x` – it has to work for at least all the `xs` that appear in `xs`! If `m` even happens to be polymorphic in `x`, then that is clearly good enough:

```
hmap :: (forall x . f x -> g x) -> NP f xs -> NP g xs
```

This is an example of a *rank-2-polymorphic* type. The argument function to `hmap` must itself be polymorphic.

If a Haskell function is polymorphic, the caller of the function has the freedom to choose the type on which to use the function, whereas the callee must make no assumptions whatsoever about the type. If an argument function is polymorphic, the situation is reversed: now the caller has the obligation to pass in a polymorphic function, and the callee may flexibly use it on any type desired.

Let us look at an example (using a `Show` instance for `NP` that we'll only define later):

```
group' :: NP I '[Char, Bool, Int]
group' = I 'x' :* I False :* I 3 :* Nil
example :: NP Maybe '[Char, Bool, Int]
example = hmap (Just . unI) group'

GHCi> example
Just 'x' :* (Just False :* (Just 3 :* Nil))
```

While we cannot map plain `Just` over `group'`, we can still get quite close. We have to unwrap the elements from the `I` constructor before we can re-wrap them in `Just`.

1.6.2 Applicative *n*-ary products

Now that we've seen `hmap`, let's see if we can come up with functions corresponding to `pure` and `<*>` as well.

Let's recall `vreplicate`:

```
vreplicate :: forall a n . SNatI n => a -> Vec a n
vreplicate x = go sNat
  where
    go :: forall n . SNat n -> Vec a n
    go SZero = VNil
    go SSuc  = x `VCons` go sNat
```

The role of `n` is now played by the signature `xs`, and the role of `a` by the type constructor `f`.

So we need two ingredients to make this work:

- A sufficiently polymorphic value of type `f x` so that it can be used for any `x` in the signature of `xs`. We'll use the type `forall a . f a` here.
- A singleton for lists.

Following the same principles we used for `SNat`, we try to define:

```
data SList (xs :: [k]) where
  SNil :: SList '[]
  SCons :: SListI xs => SList (x ': xs)
class SListI (xs :: [k]) where
  sList :: SList xs
instance SListI '[] where
  sList = SNil
instance SListI xs => SListI (x ': xs) where
  sList = SCons
```

This isn't quite ideal, because we say nothing about the elements of the list. In principle, we'd want `SCons` to include both a singleton for the head and a singleton for the tail.

For now, we can actually define:

```
hpure :: forall f xs . SListI xs => (forall a . f a) -> NP f xs
hpure x = go sList
  where
    go :: forall xs . SList xs -> NP f xs
    go SNil = Nil
    go SCons = x :* go sList
```

It is reassuring to see that the code is essentially the same as for `vreplicate`.

Here are two examples:

```
GHCi> hpure Nothing :: NP Maybe '[Char, Bool, Int]
Nothing :* (Nothing :* (Nothing :* Nil))
GHCi> hpure (K 0) :: NP (K Int) '[Char, Bool, Int]
K {unK = 0} :* (K {unK = 0} :* (K {unK = 0} :* Nil))
```

1.6.3 Lifted functions

As a final step, let us tackle `hap`, the function corresponding to `vapply`:

```
vapply :: Vec (a -> b) n -> Vec a n -> Vec b n
vapply VNil          VNil          = VNil
vapply (f `VCons` fs) (x `VCons` xs) = f x `VCons` vapply fs xs
```

It is relatively obvious that we'll have to turn this into

```
hap :: NP...xs -> NP f xs -> NP g xs
```

But what goes into the place of `...`? We need the first vector to contain at the position indexed by `x` a function of type `f x -> g x`. So the interpretation function we're looking for is something like `\x -> (f x -> g x)`, only that we do not have type-level lambda available in Haskell.

The standard trick is to define a **newtype**, which will go along with some manual wrapping and unwrapping, but is isomorphic to the desired type:

```
newtype (f -.-> g) a = Fn {apFn :: f a -> g a}
infixr 1 -.->
```

We can then define

```
hap :: NP (f -.-> g) xs -> NP f xs -> NP g xs
hap Nil Nil = Nil
hap (f :* fs) (x :* xs) = apFn f x :* hap fs xs
```

Let's look at an example:

```
lists :: NP [] '[String, Int]
lists = ["foo", "bar", "baz"] :* [1..10] :* Nil
```

```

numbers :: NP (K Int) '[String, Int]
numbers = K 2 :* K 5 :* Nil
fn_2 :: (f a -> f' a -> f'' a)
       -> (f -.-> f' -.-> f'') a
fn_2 f = Fn (\x -> Fn (\y -> f x y))
take' :: (K Int -.-> [] -.-> []) a
take' = fn_2 (\(K n) xs -> take n xs)

```

```

GHCi> hpure take' 'hap' numbers 'hap' lists
["foo", "bar"] :* ([1, 2, 3, 4, 5] :* Nil)

```

1.6.4 Another look at hmap

For normal lists, we have the identity

```
map f = pure f <*> xs
```

And indeed, the corresponding identity does also hold for environments:

```

hmap' :: SListI xs => (forall a . f a -> g a) -> NP f xs -> NP g xs
hmap' f xs = hpure (Fn f) 'hap' xs

```

is the same as `hmap`. The only difference is an additional (and rather harmless) constraint on `SListI`.

1.7 Abstracting from classes and type functions

Can we also apply `hmap` or `hmap'` to `show`, thereby turning an `NP I` into an `NP (K String)`? Unfortunately not: for example, the call

```
hmap (K . show . unI) group'
```

results in a type error, because we have

```
K . show . unI :: forall x . Show x => I x -> K String x
```

which does not match

```
forall x . f x -> g x
```

due to the class constraint on `Show`. But it would be useful to allow this, and in fact, there are other functions with other class constraints that we might also want to map over an environment, provided that all elements in the environment are actually instances of the class.

1.7.1 The kind `Constraint`

Fortunately, GHC these days also allows us to abstract over classes in this way. It turns out that classes can be seen as types with yet again a different kind. Where `data` introduces types of kind `*`, classes introduce types of kind `Constraint`. So like `*`, the kind `Constraint` is open.

Examples:

- Classes such as `Show` or `Eq` or `Ord` all have kind `* -> Constraint`. They are parameterized by a type of kind `*`, and if applied, form a class constraint.
- Classes such as `Functor` or `Monad` are of kind `(* -> *) -> Constraint`. They are parameterized by type constructors of kind `* -> *`.
- A multi-parameter type class such as `MonadReader` has kind `* -> (* -> *) -> Constraint`. It needs two parameters, one of kind `*` and one of kind `* -> *`.

In addition, we can use tuple syntax to create empty constraints and combine constraints in a syntactically somewhat ad-hoc fashion:

```
type NoConstraint      = (() :: Constraint)
type SomeConstraints a = (Eq a, Show a)
type MoreConstraints f a = (Monad f, SomeConstraints a)
```

Note that there are no such things as “nested” constraint sets – collections of constraints are automatically flattened, `(,)` acts like a union operation here.

1.7.2 Type functions

One thing we would like to express is that a constraint holds for all elements of a type-level list. It is a part of what we need in order to write a variant of `hmap` that is compatible with constrained functions.

To do this, we are going to define a *type family* – essentially a function on types, combining a parameterized constraint and a list of parameters into a single constraint:

```
type family All (c :: k -> Constraint) (xs :: [k]) :: Constraint where
  All c '[]      = ()
  All c (x ': xs) = (c x, All c xs)
```

We can use a rather cryptic command in GHCi to expand type families for us and see `All` in action:

```
GHCi> :kind! All Eq '[Int, Bool]
All Eq '[Int, Bool] :: Constraint
= (Eq Int, (Eq Bool, ()))
```

Once again, note that while GHCi shows this as a nested tuple, it is actually flattened into a single set of constraints.

We can use `All` to write a function that turns all elements in a heterogeneous list (a `HList`, not an `NP`) to strings via `show` and concatenates them:

```

hToString :: All Show xs => HList xs -> String
hToString HNil          = ""
hToString (HCons x xs) = show x ++ hToString xs

```

Let's see why this works: When pattern matching on `HCons`, GHC learns that $xs \sim y' : ys$. Therefore, we have by applying the definition of `All`:

```
All Show xs ~ All Show (y' : ys) ~ (Show y, All Show ys)
```

The call `show x` requires `Show y`, and the recursive call to `hToString xs` requires `Show ys`, so everything is fine.

1.7.3 Composing constraints

In order to do the same for `NP`, we need one extra ingredient. If we have e.g. an `NP f '[Int, Bool]`, we do not actually need

```
(Show Int, Show Bool)
```

but rather

```
(Show (f Int), Show (f Bool))
```

If we had function composition on the type level, we could also write

```
((Show 'Compose' f) Int, (Show 'Compose' f) Bool)
```

and express this as `All (Show 'Compose' f) Bool`.

In general, function composition on the type level is difficult to define in Haskell – once again because of the lack of a type-level lambda. And `newtype`-wrapping isn't helpful here, because we need to produce a `Constraint`, and a `newtype` – just like `data` – always produces something of kind `*`.

However, it turns out that for the special case of composition where the result is a `Constraint`, we can use `class` to define it:

```

class (f (g x)) => (f 'Compose' g) x
instance (f (g x)) => (f 'Compose' g) x

```

Note that the inferred kind for `Compose` is

```
Compose :: (b -> Constraint) -> (a -> b) -> a -> Constraint
```

Constraints defined via `class` can – unlike type synonyms and type families – be partially applied, so we can actually apply `Compose` as desired in connection with `All`:

```

GHCi> :kind! All (Show 'Compose' I) '[Int, Bool]
All (Show 'Compose' I) '[Int, Bool] :: Constraint
= (Compose Show I Int, (Compose Show I Bool, ()))

```

With this, we can define `hToString` for `NP`

```

hToString' :: All (Show 'Compose' f) xs => NP f xs -> String
hToString' Nil = ""
hToString' (x :* xs) = show x ++ hToString' xs

```

and try it:

```

GHCi> hToString' group'
"I {unI = 'x'}I {unI = False}I {unI = 3}"

```

This means we now also have the necessary ingredient to have GHC derive a proper `Show` instance for `NP` for us, because `show` is nothing but a more sophisticated version of our `hToString'` function with the same type requirements:

```

deriving instance (All (Compose Show f) xs) => Show (NP f xs)

```

1.7.4 Proxies

We'll now use abstraction over constraints to define a very useful variant of `hpure` that instead of taking an argument of type

```
forall a . f a
```

takes an argument of type

```
forall a . c a => f a
```

for some parameterized constraint `c`. So we're aiming for `hcpure` with a type as follows (I'm putting the type of `hpure` next to it for comparison):

```

hpure  :: SListI xs          => (forall a . f a) -> NP f xs
hcpure :: (SListI xs, All c xs) => (forall a . c a => f a) -> NP f xs

```

However, there is one more problem that we can observe without actually defining the function. Let's just say

```

hcpure :: (SListI xs, All c xs) => (forall a . c a => f a) -> NP f xs
hcpure = undefined

```

and try to call e.g.

```

GHCi> hcpure (I minBound) :: NP I '[Char, Bool]

```

Note that

```

I minBound :: Bounded a => I a

```

so it seems to have the correct type. Nevertheless, we get a type error from GHC complaining that it cannot properly match up `Bounded` with `c`. In fact, GHC generally refuses to infer constraint variables from matching available constraints, because there are situations where this can be ambiguous.

We can help GHC by providing a dummy parameter to the function – a so-called *proxy* – that has a trivial run-time representation and does nothing else but to fix the value of a type variable.

```
data Proxy (a :: k) = Proxy
```

A proxy works for arguments of any kind, so in particular for kind `*` \rightarrow `Constraint` at which we need it here. Because `data`-defined type are by construction injective, for GHC, knowing `Proxy a` is always sufficient to infer `a`.

1.7.5 A variant of `hpure` for constrained functions

So we'll actually define

```
hcpure :: forall c f xs . (SListI xs, All c xs)
      => Proxy c -> (forall a . c a => f a) -> NP f xs
hcpure p x = go sList
  where
    go :: forall xs . All c xs => SList xs -> NP f xs
    go SNil = Nil
    go SCons = x :* go sList
```

This works, as we can test:

```
GHCi> hcpure (Proxy :: Proxy Bounded) (I minBound) :: NP I '[Char, Bool]
I {unI = '\NUL'} :* (I {unI = False} :* Nil)
GHCi> hcpure (Proxy :: Proxy Show) (Fn (K . show . unI)) 'hap' group'
K {unK = "'x'"} :* (K {unK = "False"} :* (K {unK = "3"} :* Nil))
```

The second example shows that the composition of `hcpure` with the old `hap` also gives us the desired mapping of constrained functions over environments:

```
hcmep :: (SListI xs, All c xs)
      => Proxy c -> (forall a . c a => f a -> g a) -> NP f xs -> NP g xs
hcmep p f xs = hcpure p (Fn f) 'hap' xs
```


2 Generic programming with n -ary sums of products

In the first part, we've taken a tour through Haskell's type-level programming extensions, driven mostly by the desire to cover all the concepts that we need in order to understand the `generics-sop` library.

It is no surprise that we've spent a lot of time on the type `NP` of n -ary products, because we use this type to represent the fields of a constructor.

In this chapter, we'll also introduce n -ary sums, which we'll use to represent the choice between constructors. We can then explain how most Haskell datatypes can easily be transformed into sums of products.

2.1 Recap: n -ary products `NP`

Let's repeat the definition of n -ary products:

```
data NP (f :: k -> *) (xs :: [k]) where
  Nil  :: NP f '[]
  (:*) :: f x -> NP f xs -> NP f (x ': xs)
infixr 5 :*
```

The most important functions that we've considered were:

```
hpure  :: SListI xs          => (forall a . f a) -> NP f xs
hcpure :: (SListI xs, All c xs) => Proxy c -> (forall a . c a => f a) -> NP f xs
hap    ::                    NP (f -. -> g) xs -> NP f xs -> NP g xs
```

These give us a generalization of the applicative functor interface, and allow us to define other functions such as `hmap` and `hcmmap` easily:

```
hmap  :: (SListI xs)
      => (forall x . f x -> g x) -> NP f xs -> NP g xs
hcmmap :: (SListI xs, All c xs)
      => Proxy c -> (forall a . c a => f a -> g a) -> NP f xs -> NP g xs
```

We had also considered

```
hcollapse :: NP (K a) xs -> [a]
```

that can collapse a homogeneous product into a normal list.

2.2 Generalizing choice

We moved from homogeneous lists to homogeneous length-indexed vectors, then to heterogeneous lists and ultimately to `NP`.

2.2.1 Choosing from a list

We'll try to go through a similar progression for choices or sums. A Haskell list is a sequence of arbitrarily many elements of type `a`. Let's also define a type that is a choice between arbitrarily many elements of type `a`:

```
data LChoice a = LCZero a | LCSuc (LChoice a)
```

If we imagine arbitrarily elements of type `a` being represented by a list of `as`, then `LChoice a` gives us the index of a particular element in that list paired with the element being chosen. The constructors are named `LCZero` and `LCSuc` so that we can view an easily view values of type `LChoice a` as a zero-based index, where `LCZero` contains the payload.

Just to make the relation to the following datatypes more obvious, let's rewrite `LChoice` equivalently in GADT syntax:

```
data LChoice (a :: *) where
  LCZero :: a -> LChoice a
  LCSuc  :: LChoice a -> LChoice a
```

2.2.2 Choosing from a vector

If we now generalize from lists to vectors, we'll get a choice that is additionally indexed by the number of elements that we can choose from

```
data VChoice (a :: *) (n :: Nat) where
  VCZero :: a -> VChoice a (Suc n)
  VCSuc  :: VChoice a n -> VChoice a (Suc n)
```

We can choose the first element (via `VCZero`) from any non-empty vector – so `VCZero` forces the length index to be of form `Suc n`.

If we have an index into an `n`-element vector, we can also produce an index into an `Suc n`-element vector by skipping the first element. This is what `VCSuc` does. Once again, we pair the choice with the actual element of type `a` being chosen.

Note that there are no possible choices of an `a` from an empty vector, which is reflected by the fact that there are no constructors of `VChoice` targeting the index `Zero`.

2.2.3 Choosing from a heterogeneous list

The next step is to generalize from vectors to `HList`, so we'll have an index that is a list of types:

```
data HChoice (xs :: [*]) where
  HCZero :: x -> HChoice (x ': xs)
  HCSuc  :: HChoice xs -> HChoice (x ': xs)
```

Now the element chosen, which is still stored in `HCZero`, is taken from the index list. In analogy with `VChoice`, we have no constructors targeting the empty index list.

2.2.4 Choosing from an environment

For n -ary products, we decided that it's additionally helpful to build type constructor application into the type. We'll perform the same generalization for our choice, and arrive at our type of n -ary sums:

```
data NS (f :: k -> *) (xs :: [k]) where
  Z :: f x -> NS f (x ': xs)
  S :: NS f xs -> NS f (x ': xs)
```

The payload is now of type $f\ x$, where x is from the index list. Just as for `NP`, we can actually be kind-polymorphic in the index list, because only elements of type $f\ x$ directly appear as data in the structure.

Once again, if we choose $f = I$, the type is equivalent to `HChoice`, and if we choose $f = K\ a$, we get something that is much like `VChoice`, where the elements in the index list are ignored, and just the length of it determines the number of as we can choose from.

Let's look at an example. Let's consider the type

```
type ExampleChoice = NS I '[Char, Bool, Int]
```

So we can choose a character with index 0, a Boolean with index 1, or an integer with index 2:

```
c0, c1, c2 :: ExampleChoice
c0 = Z (I 'x')
c1 = S (Z (I True))
c2 = S (S (Z (I 3)))
```

It's useful for experimenting if we have a `Show` instance for `NS` available. We can let GHC derive one in much the same way as we have for `NP` – we require that all elements that can appear in the `NS` have show instances themselves:

```
deriving instance All (Show 'Compose' f) xs => Show (NS f xs)
```

2.3 Sums of products

Having defined both `NS` and `NP`, we are now at a point where we can define the uniform generic representation that we are going to use for datatypes.

2.3.1 Representing expressions

Let's recap from the very beginning. If we have a datatype such as e.g.

```
data Expr = NumLit Int | BoolLit Bool | Add Expr Expr | If Expr Expr Expr
```

then each value of that type is of form

$$C_i\ x_0 \dots x_{n_i-1}$$

where C_i is one of the constructors, in this case either `NumLit`, `BoolLit`, `Add` or `If`. And the `xs` are the arguments of the chosen constructor.

We use `NS` to represent the choice between the constructors, and we have four things to choose from. We use `NP` to represent the sequence of arguments. The products contain actual elements of the argument types. So the representation type corresponding to `Expr` is

```
type RepExpr = NS (NP I) (' ['[Int]
                          , '[Bool]
                          , '[Expr, Expr]
                          , '[Expr, Expr, Expr]
                          ])
```

The index is now a list of list of types – a table that, except for the names of the datatype and the constructors, completely specifies the structure of the datatype. The outer list reflects the sum structure. Each list element corresponds to one constructor. The inner lists reflect the product structures, and each element corresponds to one argument.

Note that we only represent the top-level structure of an expression. A `RepExpr` contains values of the original type `Expr`. We never look at the arguments of a constructor further. Conversion between a datatype and its structural representation will always be shallow and non-recursive. If needed, we can and will apply such conversions multiple times while traversing a recursive structure.

Let's consider an example value of type `Expr`:

```
exampleExpr :: Expr
exampleExpr = If (BoolLit True) (NumLit 1) (NumLit 0)
```

The corresponding `RepExpr` is:

```
exampleRepExpr :: RepExpr
exampleRepExpr = S (S (S (Z (I (BoolLit True) :* I (NumLit 1) :* I (NumLit 0) :* Nil))))
```

In the following, we're sometimes going to use a more concise syntax in order to represent values of types `NS` and `NP`:

```
exampleRepExpr :: RepExpr
exampleRepExpr = C3 [I (BoolLit True), I (NumLit 1), I (NumLit 0)]
```

I.e., we'll collapse the sequence of constructor applications `S (S (Z...))` into C_2 and use ordinary list syntax for the n -ary product.

It is easy to see that `RepExpr` is isomorphic to `Expr`:

```
fromExpr :: Expr -> RepExpr
fromExpr (NumLit n) = Z (I n :* Nil)
fromExpr (BoolLit b) = S (Z (I b :* Nil))
fromExpr (Add e1 e2) = S (S (Z (I e1 :* I e2 :* Nil)))
fromExpr (If e1 e2 e3) = S (S (S (Z (I e1 :* I e2 :* I e3 :* Nil))))

toExpr :: RepExpr -> Expr
toExpr (Z (I n :* Nil)) = NumLit n
toExpr (S (Z (I b :* Nil))) = BoolLit b
toExpr (S (S (Z (I e1 :* I e2 :* Nil)))) = Add e1 e2
toExpr (S (S (S (Z (I e1 :* I e2 :* I e3 :* Nil)))))) = If e1 e2 e3
```

2.3.2 The `Generic` class

In the beginning, we presented the following class `Generic` to capture the idea of datatype-generic programming:

```
class Generic a where
  type Rep a
  from :: a -> Rep a
  to   :: Rep a -> a
```

We're now at a point where we can define the actual class `Generic` that is used in `generics-sop`. Because all representations will be of the form

```
NS (NP I) c
```

for some code `c :: [[*]]`, we decide to define a `newtype` and a type synonym:

```
newtype SOP f a = SOP (NS (NP f) a)
type Rep a = SOP I (Code a)
```

and include `Code` in the class rather than `Rep`. Additionally, we're going to demand that both the outer list and all inner lists of the code have implicit singletons available. This isn't strictly speaking necessary, but it's harmless and will reduce the number of constraints we need to put on function definitions later:

```
class (SListI (Code a), All SListI (Code a)) => Generic (a :: *) where
  type Code a :: [[*]]
  from :: a -> Rep a
  to   :: Rep a -> a
```

The instantiation of `Generic` to e.g. `Expr` is straight-forward and mechanical:

```
instance Generic Expr where
  type Code Expr = ' [ '[Int]
                    , '[Bool]
                    , '[Expr, Expr]
                    , '[Expr, Expr, Expr]
                    ]
  from x = SOP (fromExpr x)
  to (SOP x) = toExpr x
```

2.3.3 Other `Generic` instances

The presence of datatype parameters does not make the definition of a `Generic` instance harder, and we can easily provide `Generic` instances even for built-in types such as Haskell lists:

```
instance Generic [a] where
  type Code [a] = ' [' [], '[a, [a]]
```

```

from []      = SOP (Z Nil)
from (x : xs) = SOP (S (Z (I x :* I xs :* Nil)))
to (SOP (Z Nil)) = []
to (SOP (S (Z (I x :* I xs :* Nil)))) = x : xs

```

The `Generic` instances are the boilerplate we still have to write in order to be able to benefit from the `generics-sop` approach. We'll see that once we have a `Generic` instance, we get all generic functions we write for free for all these types.

But the `Generic` instances themselves are easy to automate. In practice, we have a number of options:

- The first and generally worst option is to write `Generic` instances by hand. This is quickly getting tedious, but in principle gives us more flexibility, because we could decide to define non-standard `Generic` instances.
- The second is to use Template Haskell to derive `Generic` instances for us. This is easy to do, and provided by the `generics-sop` library. We can just put

```
deriveGeneric ''Expr
```

after the `data` declaration of `Expr` and get the `Generic` instance derived for us automatically.

- The third option is to extend the compiler with support to derive instances of `Generic`, just like it can derive instances of `Eq` and `Show`. Then we could simply write

```

data Expr = ...
  deriving (... , Generic)

```

and would also get the instance for us automatically. This option deserves a bit more discussion.

2.3.4 Support for `Generic` in the compiler

As discussed in the beginning, there are many – somewhat competing – approaches to generic programming in Haskell, which use slightly different views on the datatypes. While it is conceivable that GHC supports one or even two of these directly, it's not feasible to support all of them directly. Each new class that has built-in `deriving` requires a compiler change and subsequent maintenance of the feature.

In fact, GHC has first-class support for two generic programming approaches already. For SYB, it allows `deriving Data`, and for generic-deriving, it allows `deriving Generic` – but for a different class also named `Generic`.

Now, an interesting realization is that with all the type-level programming features that GHC offers and we have discussed, we actually have sufficient machinery to turn the representations GHC has built-in support for into other generic representations programmatically! This makes extensive use of type classes and typ families that turn one structural representation into another, and at the same time provide functions that convert values from one representation into values in another (and back).

It is e.g. possible to convert from GHC’s own `Generic` class into the `generics-sop`-provided `Generic` class using this technique. This is also implemented in the `generics-sop` library. To use this approach, we have to first import GHC’s `Generic` class in qualified form:

```
import qualified GHC.Generics as GHC
```

Then, when defining a datatype such as `Expr`, we tell GHC to derive its own `Generic` class for this type:

```
data Expr = ...
  deriving (... , GHC.Generic)
```

Finally, we can just provide an empty instance for `generics-sop`’s `Generic` class:

```
instance Generic Expr
```

The implementation of `generics-sop` has default definitions for `Generic` that will use and translate from `GHC.Generic`.

This technique of translating different generic programming representations into each other that we have dubbed “Generic Generic Programming” has a significant impact: GHC should no longer strive to provide built-in support for the “one true” or “best” approach to generic programming, nor do we need a “one size fits all” approach to generic programming. GHC instead should provide support for a representation that captures as much information as possible, even if that means that the representation is difficult to use. But it means we can translate easily from that representation to various others, forgetting unnecessary things along the way. Generic programming libraries also no longer have to compete for the one or two spots of first-class support, but can all co-exist with each other and be used together by users in one and the same program.

2.4 Defining generic equality

Let us now look at defining one standard example of datatype-generic programming in our approach – generic equality.

2.4.1 A bottom-up approach

Let’s start bottom-up, by comparing two n -ary products:

```
geqNP :: All Eq xs => NP I xs -> NP I xs -> Bool
geqNP Nil Nil = True
geqNP (I x1 :* xs1) (I x2 :* xs2) = x1 == x2 && geqNP xs1 xs2
```

This is rather straight-forward. We traverse both products in parallel, and compare all the contained elements for equality pointwise. We fall back to the `Eq` class for the elements, which means we have to require an `Eq` instance for all types that occur in the signature, using `All`.

(In fact, we can do this in a slightly more high-level way, and will return to this point in a moment.)

Next, lets' try to compare to n -ary sums of n -ary products:

```
geqNS :: ... => NS (NP I) xss -> NS (NP I) xss -> Bool
geqNS (Z np1) (Z np2) = geqNP np1 np2
geqNS (S ns1) (S ns2) = geqNS ns1 ns2
geqNS _ _ = False
```

The idea is also clear. We traverse the two sums and check whether both are mapping to the same choice. If it's obvious that they don't, we return `False`. Otherwise, we call `geqNP` on the products contained within.

2.4.2 The `All2` constraint

The problem here is that due to `geqNP`, we need a constraint that all elements of all the *inner* lists in the list of lists `xss` are instances of `Eq`. If we had partial parameterization of type families available, this would be easy to do:

```
All (All Eq) xss
```

However, GHC does not admit that. Instead, we simply define a separate type family `All2` that works on lists of lists. This is not nice, but neither is it tragic, because due to the two-layer structure of our datatypes `All` and `All2` are the only cases we'll ever need.

```
type family All2 (f :: k -> Constraint) (xss :: [[k]]) :: Constraint where
  All2 f '[] = ()
  All2 f (xs ': xss) = (All f xs, All2 f xss)
```

With `All2`, we can now complete `geqNS`:

```
geqNS :: All2 Eq xss => NS (NP I) xss -> NS (NP I) xss -> Bool
geqNS (Z np1) (Z np2) = geqNP np1 np2
geqNS (S ns1) (S ns2) = geqNS ns1 ns2
geqNS _ _ = False
```

2.4.3 Completing the definition

From here, we can go to `geqSOP` simply by stripping the `SOP` constructors:

```
geqSOP :: All2 Eq xss => SOP I xss -> SOP I xss -> Bool
geqSOP (SOP sop1) (SOP sop2) = geqNS sop1 sop2
```

And now, it is easy to define a generic equality that works for all representable types:

```
geq :: (Generic a, All2 Eq (Code a)) => a -> a -> Bool
geq x y = geqSOP (from x) (from y)
```

If we now wanted to use `geq` to define equality for a datatype, such as `Expr`, we could do so instead of invoking `deriving Eq` by giving a trivial instance declaration:


```
instance Eq Expr where
  (==) = geq
```

Let's try this:

```
GHCi> geq (Add (NumLit 3) (NumLit 5)) (Add (NumLit 3) (NumLit 5))
True
GHCi> geq (Add (NumLit 3) (NumLit 5)) (Add (NumLit 3) (NumLit 6))
False
```

2.4.4 Improving the product case

Not without reason, we spent a lot of time defining useful functions over `NP`. One of the advantages of the generics-sop approach is that it encourages defining high-level operations that work on `NP` and `NS`, and then using a composition of such operators to define (parts of) generic functions.

Let's look again at `geqNP`:

```
geqNP :: All Eq xs => NP I xs -> NP I xs -> Bool
geqNP Nil Nil = True
geqNP (I x1 :* xs1) (I x2 :* xs2) = x1 == x2 && geqNP xs1 xs2
```

We can see this as a composition of several different operations: first, we zip the two products together using `(==)`, getting a homogeneous product of `Bools`. We can thus collapse that product into a `[Bool]` and then just fold with `(&&)` and `True` over that list.

The zipping we need here turns out to be a straight-forward generalization of `hcmmap`, again easy to define using `hcpure` and `hap`:

```
hzipWith :: (All c xs, SListI xs)
=> Proxy c
-> (forall a . (c a) => f a -> g a -> h a)
-> NP f xs -> NP g xs -> NP h xs
hzipWith p f xs1 xs2 = hcpure p (fn_2 f) 'hap' xs1 'hap' xs2
```

We can now redefine `geqNP` as follows:

```
geqNP' :: (SListI xs, All Eq xs) => NP I xs -> NP I xs -> Bool
geqNP' xs ys = and
  $ hcollapse
  $ hzipWith (Proxy :: Proxy Eq) aux xs ys
where
  aux (I x) (I y) = K (x == y)
```

The only tiny annoyance is that we have to unwrap and rewrap the arguments of `(==)` from with `I` and `K` constructors as appropriate. We also pay the price of an additional `SListI xs` constraint that we then also have to add to the other functions that are used to define `geq` – but ultimately, it does not change anything, because we had already imposed `SListI` constraints as superclass constraints on `Generic`.